Math 2280-2
Example 2.3.1--2.3.3.

A crossbow bolt is shot upwards with velocity v0 from the earth.
What is the maximum height it reaches?
How long does it stay aloft?
Define some constants (initial position, velocity and gravitational acceleration

```
> y0:=0: v0:=49: g:=9.8:
```

Case 1: Without air resistance (drag)

Define the velocity and position

```
> v1:=t->-g*t+v0; y1:=t->-g/2*t^2+v0*t+y0;
```

$$v1 := t \to -g\,t + v0$$

$$y1 := t \to -\frac{1}{2}\,g\,t^2 + v0\,t + y0 \qquad \textbf{(1.1)}$$

Max height is when v1(t)=0

```
> t1max:=v0/g; y1max:=y1(t1max);
```

$$t1max := 5.000000000$$

$$y1max := 122.5000000 \qquad \textbf{(1.2)}$$

The object lands when y1(t)=0

```
> t1landsols:=solve(y1(t)=0,{t});
  t1land:=subs(t1landsols[2],t); # here we select and extract
  the second sol
```

$$t1landsols := \{t = 0.\},\ \{t = 10.\}$$

$$t1land := 10. \qquad \textbf{(1.3)}$$

Case 2: With linear air resistance (drag proportional to velocity)

Here we assume we know the terminal velocity is

```
> vt:=-245;
```

$$vt := -245 \qquad \textbf{(2.1)}$$

Thus the drag coefficient is:

```
> rho:=g/(-vt);
```

$$\rho := 0.04000000000 \qquad \textbf{(2.2)}$$

The velocity and position are given by

```
> v2:=t->vt+(v0-vt)*exp(-rho*t);
  y2:=t->y0+vt*t + (1/rho)*(v0-vt)*(1-exp(-rho*t));
```

$$v2 := t \to vt + (v0 - vt)\,e^{-\rho\,t}$$

$$y2 := t \rightarrow y0 + vt\,t + \frac{(v0 - vt)\left(1 - e^{-\rho t}\right)}{\rho} \qquad (2.3)$$

Max height is when v2(t)=0

```
> t2maxsols:=solve(v2(t)=0,{t});
  tmax:=subs(t2maxsols,t);
  y2max:=y2(tmax);
```

$$t2maxsols := \{t = 4.558038920\}$$

$$tmax := 4.558038920$$

$$y2max := 108.280465 \qquad (2.4)$$

The bolt lands when y2(t)=0

```
> t2landsols:=solve(y2(t)=0,{t});
  t2land:=subs(t2landsols[1],t); # here we select and extract
  the first sol
```

$$t2landsols := \{t = 9.410949931\},\ \{t = 0.\}$$

$$t2land := 9.410949931 \qquad (2.5)$$

And the landing velocity is

```
> v2land:=v2(t2land);
```

$$v2land := -43.2273093 \qquad (2.6)$$

Now let us compare both trajectories

```
> plot({y1(t),y2(t)},t=0..10,color=black);
```



Case 3: With quadratic air resistance

Here we follow the discussion in pp103-105 of book. It is natural to define a different drag coefficient:

```
> rho2:=0.0011;
```

$$\rho 2 := 0.0011 \quad \textbf{(3.1)}$$

The velocity is a piecewise constant function (depends on whether we go up or down)

```
> C1:=arctan(v0*sqrt(rho2/g)); C2:=arctanh(v0*sqrt(rho2/g));
  tmax1:=C1/sqrt(rho2*g); tmax2:=C2/sqrt(rho2*g);
  v3:=t->piecewise(t<tmax1,sqrt(g/rho2)*tan(C1-t*sqrt(rho2*g)),
  t>=tmax1,sqrt(g/rho2)*tanh(C2-(tmax2+t-tmax1)*sqrt(rho2*g)));
  y3:=t->int(v3(s),s=0..t);
```

$$C1 := 0.4788372920$$

$$C2 := 0.5751533900$$

$$tmax1 := 4.611886235$$

$$tmax2 := 5.539547664$$

$$v3 := t \rightarrow piecewise\left(t < tmax1, \sqrt{\frac{g}{\rho 2}} \tan\left(C1 - t\sqrt{\rho 2\, g}\right), tmax1 \leq t, \sqrt{\frac{g}{\rho 2}} \tanh\left(C2 - (tmax2 + t - tmax1)\sqrt{\rho 2\, g}\right)\right)$$

$$y3 := t \rightarrow \int_0^t v3(s)\, ds \qquad \textbf{(3.2)}$$

Now we compare all three drag models. Notice that the linear and quadratic drags give almost identical results!

```
> plot(v3(t),t=0..10,color=black,title="velocity");
  plot({y1(t),y2(t),y3(t)},t=0..10,color=black,title=
  "comparison of position for three drag models");
```




comparison of position for three drag models
120
100
80
60
40
20
0
-20
2
4
6
8
10
t

>